Slides for Chapter 6: General Equilibrium with Distortionary Taxes, Public Goods, Externalities, Optimal Taxation and Redistribution policies

6.1 Taxes in the benchmark equilibrium

A positive tax and tax revenue are present in the benchmark data. The first task: construct a micro-consistent data set. *Remember that entries are values*.

Each tax should be added as a row to the matrix.

Taxes are negative entries in a column indicating payments by a sector.

There is a corresponding positive entry somewhere. In the present case, the tax is redistributed lump sum to the consumer, so the consumer gets a positive entry of the tax revenue.

A zero row sum for the tax indicates that all tax receipts must be paid to someone.

	Production Sectors			Consumers
Markets	X	Y	W	CONS
PX	100		-100	
PY		100	-100	
PW			200	-200
PL	-20	-60		80
PK	-60	-40		100
TAX	-20	0		20

X sector receives 100 units of revenue, of which 20 is paid in taxes.

This 20 is received as part of the consumer’s income.

These data do not indicate what type of tax is in place. It could be a tax on X output, on all the inputs, or on just one input. We interpret this as a tax on the labor input into sector X.

A crucial task: keep track of what prices firms and consumers face. It is (generally) not possible to calibrate a benchmark equilibrium with all prices equal to one.

If a production input is taxed, then if its consumer price (price received by the consumer) is chosen to be equal to one, then producer price (price paid by the producer) is specified as $(1 + t)$.

If the producer price is unity, the consumer price is $1/(1+t)$.

Given that we interpret the above data as a tax on the labor input into the X sector, the data tell us that the tax rate is 100%.

The amount paid by the X sector to labor (20) is equal to the tax revenue (20).

Thus if we set the consumer price of labor to 1 (also the price to the Y sector), then the price of labor to the X sector must be 2.

Allow for alternative taxes in the model, including a tax on capital inputs into X (TKX) and a tax on X output (TX), set to zero initially.

```
100*((1+TLX)*PL/2 )**0.4 * (PK*(1+TKX))**0.6

           =G= 100*PX*(1-TX);
```

Counterfactual: eliminate taxes on X sector inputs and replace with a single tax on X sector output. Then taxes on both inputs.

The output tax rate will be different from the corresponding tax rate on all inputs, because the tax base is different in the two cases.

Let *mc* denote the marginal cost of production (or producer price) and *p* denote the price charged to the consumer. This is how MPS/GE interprets input (*ti*) versus output (*to*) taxes.

Tax on all inputs: $p = (1 + ti)mc$
Tax on the output: $p(1 - to) = mc$

Note *mc* is the tax base for the input tax, and *p* is the tax base for the output tax. The output tax that is equivalent to the tax on all inputs is found by:

$$(1 + ti) = 1/(1 - to)$$

If *ti* = TLX = TKX = 0.25 as we have assumed in our first counterfactual, then the equivalent output tax is given by *to* = TX = 0.20.

One more equivalence: The final counterfactual demonstrates that a 20% tax on the output of X is the same as a 25% subsidy to the production of Y.

Let *t* be the tax on X and *s* the subsidy to Y. Formally, we have

$$\frac{p_x(1 - t)}{p_y} = \frac{p_x}{p_y(1 + s)} = \frac{mc_x}{mc_y} \qquad if \quad t = 0.20, \quad s = 0.25$$

Absolute prices may differ depending on the choice of the numeraire, but all quantities and welfare are the same.

```
$TITLE Model M6-1: 2x2 (two goods, two factors) benchmark taxes
*  Positive tax in the X sector in the benchmark

$ONTEXT
                Production Sectors          Consumers

   Markets   |    X        Y         W    |        CONS
   ------------------------------------------------------
        PX   |  100               -100    |
        PY   |          100       -100    |
        PW   |                     200    |        -200
        PL   |  -20     -60               |          80
        PK   |  -60     -40               |         100
        TAX  |  -20       0               |          20
   ------------------------------------------------------

Assume that this is a 100% tax on labor in X: TLX = 1.
Let the CONSUMER price (wage) of labor equal 1: PL = 1.
The PRODUCER price (cost) of labor in X is equal to 2:
   PL*(1+TLX) = 2

$OFFTEXT

SCALAR  TX      Proportional output tax on sector X,
        TY      Proportional output tax on sector Y,
        TLX     Ad-valorem tax on labor inputs to X,
```

```
        TKX      Ad-valorem tax on capital inputs to X
        TAXREV   Total tax revenue from all sources;

POSITIVE VARIABLES

        X        Activity level for sector X
        Y        Activity level for sector Y
        W        Activity level for sector W
        PX       Price index for commodity X
        PY       Price index for commodity Y
        PL       Price index for primary factor L
        PK       Price index for primary factor K
        PW       Price index for welfare (expenditure function)
        CONS     Income definition for CONS
        PPLX     Producer price for L in X
        PPKX     Producer price for K in X
        PPX      Producer price for X
        PPY      Producer price for Y;

EQUATIONS

        PRF_X    Zero profit for sector X
        PRF_Y    Zero profit for sector Y
        PRF_W    Zero profit for sector W

        MKT_X    Supply-demand balance for commodity X
```

```
        MKT_Y   Supply-demand balance for commodity Y
        MKT_L   Supply-demand balance for primary factor L
        MKT_K   Supply-demand balance for primary factor L
        MKT_W   Supply-demand balance for aggregate demand

        I_CONS  Income definition for CONS

        RPPLX   Relation between consumer and producer price L in X
        RPPKX   Relation between consumer and producer price K in X
        RPPX    Relationship between producer and consumer price of X
        RPPY    Relationship between producer and consumer price of Y;

*       Zero profit conditions:

PRF_X..  100*(PPLX/2)**0.4 * (PPKX)**0.6 =G= 100*PPX;

PRF_Y..  100*PL**0.6 * PK**0.4 =G= 100*PPY;

PRF_W..  200*PX**0.5 * PY**0.5 =G= 200*PW;

*       Market clearing conditions:

MKT_X..  100*X =G= 100*W*PW/PX;

MKT_Y..  100*Y =G= 100*W*PW/PY;
```

```
MKT_W..  200*W =G= CONS/PW;

MKT_L..  80  =G= 20*X*PPX/(PPLX/2) + 60*Y*PPY/PL;

MKT_K..  100 =G= 60*X*PPX/PPKX     + 40*Y*PPY/PK;

*       Income constraints:

I_CONS.. CONS =E= 80*PL + 100*PK +  100*PX*X*TX + 100*PY*Y*TY +
                  TLX*PL*20*X* PPX /(PPLX/2) +
                  TKX*PK*60*X* PPX /(PPKX);

RPPLX..  PPLX =E= PL*(1+TLX);
RPPKX..  PPKX =E= PK*(1+TKX);
RPPX..   PPX  =E= PX*(1-TX);
RPPY..   PPY  =E= PY*(1-TY);

MODEL BENCHTAX /PRF_X.X, PRF_Y.Y, PRF_W.W,
                  MKT_X.PX, MKT_Y.PY, MKT_L.PL, MKT_K.PK,
                  MKT_W.PW, I_CONS.CONS,
                  RPPLX.PPLX, RPPKX.PPKX,RPPX.PPX, RPPY.PPY /;

X.L     =1;
Y.L     =1;
W.L     =1;
```

```
PL.L    =1;
PX.L    =1;
PY.L    =1;
PK.L    =1;
PW.FX   =1;
PPLX.L  = 2;
PPKX.L  = 1;
PPX.L   = 1;
PPY.L   = 1;

CONS.L  =200;

TX      =0;
TY      =0;
TLX     =1;
TKX     =0;

BENCHTAX.ITERLIM = 0;
SOLVE BENCHTAX USING MCP;

BENCHTAX.ITERLIM = 1000;
SOLVE BENCHTAX USING MCP;

TAXREV = 100*PX.L*X.L*TX + 100*PY.L*Y.L*TY +
          TLX*PL.L*20*X.L* PPX.L /(PPLX.L/2) +
          TKX*PK.L*60*X.L* PPX.L /(PPKX.L);
```

```
DISPLAY TAXREV;

*   In the first counterfactual, we replace the tax on
*   labor inputs by a uniform tax on both factors:

TLX = 0.25;
TKX = 0.25;
TX  = 0;
TY  = 0;

SOLVE BENCHTAX USING MCP;

TAXREV = 100*PX.L*X.L*TX + 100*PY.L*Y.L*TY +
          TLX*PL.L*20*X.L* PPX.L /(PPLX.L/2) +
          TKX*PK.L*60*X.L* PPX.L /(PPKX.L);
DISPLAY TAXREV;

*    Now demonstrate that a 25% tax on all inputs
*    is equivalent to a
*    20% tax on the output (or all outputs if more than one)

TLX = 0;
TKX = 0;
TX  = 0.2;
TY  = 0;
```

```
SOLVE BENCHTAX USING MCP;

TAXREV = 100*PX.L*X.L*TX + 100*PY.L*Y.L*TY +
          TLX*PL.L*20*X.L* PPX.L /(PPLX.L/2) +
          TKX*PK.L*60*X.L* PPX.L /(PPKX.L);
DISPLAY TAXREV;

*    Demonstrate that a 20% tax on the X sector output is
*    equivalent to a 25% subsidy on Y sector output
*    (assumes that the funds for the subsidy can be raised
*    lump sum from the consumer!)

TKX = 0;
TLX = 0;
TX = 0;
TY = -0.25;

SOLVE BENCHTAX USING MCP;
TAXREV = 100*PX.L*X.L*TX + 100*PY.L*Y.L*TY +
          TLX*PL.L*20*X.L* PPX.L /(PPLX.L/2) +
          TKX*PK.L*60*X.L* PPX.L /(PPKX.L);
DISPLAY TAXREV;

*    Show welfare under non-distortionary taxation

TX = 0.20;
```

```
TY = 0.20;

SOLVE BENCHTAX USING MCP;
TAXREV = 100*PX.L*X.L*TX + 100*PY.L*Y.L*TY +
          TLX*PL.L*20*X.L* PPX.L /(PPLX.L/2) +
          TKX*PK.L*60*X.L* PPX.L /(PPKX.L);
DISPLAY TAXREV;

TX = 0.0;
TY = 0.0;

SOLVE BENCHTAX USING MCP;
TAXREV = 100*PX.L*X.L*TX + 100*PY.L*Y.L*TY +
          TLX*PL.L*20*X.L* PPX.L /(PPLX.L/2) +
          TKX*PK.L*60*X.L* PPX.L /(PPKX.L);
DISPLAY TAXREV;
```

6.2a Labor supply and labor tax

This model is an extension of the previous model and also extends our earlier model with endogenous labor supply (M3-6) to a case with taxes in the benchmark.

	Production Sectors					Consumers
Markets	A	B	W	TL	TK	CONS
PX	120		-120			
PY		120	-120			
PW			340			-340
PLS	-48	-72		120		
PKS	-72	-48			120	
PL			-100	-80		180
PK					-80	80
TAX				-40	-40	80

There are supply activities for labor (TL) and capital (TK). Labor can also be used for leisure and so the activity level for labor supply will vary.

Capital has no alternative use so it will always be completely supplied to the market.

Still, it can be convenient to specify a supply activity, since there will be two prices, one the consumer price and one the producer price (user cost) of capital.

Our choice will be that the consumer prices (prices received by the consumer) for labor and capital will be set to one.

The data matrix indicates that there is a 50% tax on each factor in the benchmark, so the producer prices (user costs) of labor and capital will be PLS = PKS = 1.5.

We can also choose how to interpret the X and Y values, but there is only a single price for both producers and consumers, so we will interpret these as 120 units at a price of 1 for each

One useful trick for checking the calibration and noting which sectors or markets are out of balance is to not allow the model to iterate initially.

```
PLS.L =1.5; PKS.L =1.5;

INCOMETAX.ITERLIM = 0;
SOLVE INCOMETAX USING MCP;

M32.ITERLIM = 2000;
SOLVE INCOMETAX USING MCP;
```

Suppose that we had set the initial value of PLS.L = 1.0 instead of 1.5. Look at the listing file.

	LOWER	LEVEL	UPPER	MARGINAL
---- VAR X	.	1.000	+INF	-8.440
---- VAR Y	.	1.000	+INF	-12.435
---- VAR W	.	1.000	+INF	.
---- VAR TL	.	1.000	+INF	20.000
---- VAR TK	.	1.000	+INF	.
---- VAR PX	.	1.000	+INF	.
---- VAR PY	.	1.000	+INF	.
---- VAR PL	.	1.000	+INF	.
---- VAR PK	.	1.000	+INF	.
---- VAR PLS	.	1.000	+INF	-9.163
---- VAR PKS	.	1.200	+INF	8.365
---- VAR PW	1.000	1.000	1.000	EPS
---- VAR CONS	.	340.000	+INF	.

The model has not solved. Recall from chapter 1 that GAMS writes inequalities in the greater-than-or-equal-to format.

The MARGINAL column of the listing file gives the degree of imbalance in an inequality, left-hand side minus right-hand side.

A positive number is ok if the associated variable is zero, as in a cost equation (marginal cost minus price is positive if associated with a slack activity).

A *negative* value of a marginal *cannot* be an equilibrium; for an activity it indicates positive profits and for a market it indicates demand exceeds supply.

In our incorrect calibration in which we give the producer price of labor too low a value, we see that there are positive profits for X, Y and negative profits for labor supply. There is an excess demand for labor and an excess supply for capital.

Most calibration errors are in the MPS/GE file itself, and not just in setting the initial values of the variables.

You could work with this file as an exercise, deliberately introducing errors (such as in the price fields) and see what happens. In any case, the iterlim = 0 statement is very useful in helping you identify where the errors are.

The other useful feature we introduce in this model is the use of the LOOP statement to simplify the repeated solving of the model over a series of parameter values. Two parameters are declared as vectors, WELFARE(S), and LABSUP(S) (for labor supply).

Then the loop statement sets the taxes at different values over the values of the set.

```
LOOP(S,

TXL = 0.60 - 0.10*ORD(S);
TXK = 0.40 + 0.10*ORD(S);

SOLVE ALGEBRAIC USING MCP;

WELFARE(S) = W.L;
LABSUP(S) = TL.L;
INCOME(S) = ((PX.L/1.5)*X.L + (PY.L/1.5)*Y.L)
           /((PX.L/1.5)**0.5*(PY.L/1.5)**0.5)/2;
CAPTAX(S) = TXK;
TAXREV(S) = (TXL*PL.L*TL.L*80 + TXK*PK.L*TK.L*80)
           /((PX.L/1.5)**0.5*(PY.L/1.5)**0.5);
);

DISPLAY WELFARE, LABSUP, INCOME, CAPTAX, TAXREV;
```

ORD(S) denotes the ordinal value of a member of a set. S is an indicator and is not treated as a number in GAMS, so 0.05*S won't work.

ORD(S) is treated as a number, so this is how the set index is translated into a number. Note from the tax assignment statement that when S = 1, the initial values of both taxes are 0.20, our benchmark values. At S = 5, the values are TXL = 0, and TXK = 0.40.

The model is repeatedly solved within the loop, and after each solve statement the value of the parameters WELFARE and LABSUP are assigned values. The loop is closed with "); "

After the loop is closed we ask GAMS to display the parameters at the end of the listing file. Note the set index for the parameters is not used in the display statement, GAMS knows what it is.

```
$TITLE  M6-2a.GMS: 2x2 Economy with labor supply and income tax

$ONTEXT
                  Production Sectors                Consumers

   Markets    |    X       Y        W       TL    TK       CONS
   ----------------------------------------------------------
        PX    |  120              -120
        PY    |          120      -120
        PW    |                    340                     -340
        PLS   |  -48     -72               120
        PKS   |  -72     -48                     120
        PL    |                   -100     -80              180
        PK    |                                  -80         80
        TAX   |                            -40   -40         80
   ----------------------------------------------------------
$OFFTEXT

SETS S /1*6/;

PARAMETERS
  TXL          Labor income tax rate,
  TXK          Capital income tax rate,
  WELFARE(S)   Welfare,
  LABSUP(S)    Labor supply
  INCOME(S)    Money income = consumption of X and Y
```

```
  CAPTAX(S)    The level of the capital tax
  TAXREV(S)    Tax revenue generated;

POSITIVE VARIABLES

        X          Activity level for sector X
        Y          Activity level for sector Y
        TL         Supply activity for L
        TK         Supply activity for K
        W          Activity level for sector W

        PX         Price index for commodity X
        PY         Price index for commodity Y
        PL         Price index for primary factor L net of tax
        PK         Price index for primary factor K net of tax
        PLS        Price index for primary factor L gross of tax
        PKS        Price index for primary factor K gross of tax
        PW         Price index for welfare (expenditure function)

        CONS       Income definition for CONS;

EQUATIONS

        PRF_X      Zero profit for sector X
        PRF_Y      Zero profit for sector Y
        PRF_TL     Zero profit for sector TL
```

```
        PRF_TK  Zero profit for sector TK
        PRF_W   Zero profit for sector W

        MKT_X   Supply-demand balance for commodity X
        MKT_TK  Supply-demand balance for commodity TK
        MKT_TL  Supply-demand balance for commodity TL
        MKT_Y   Supply-demand balance for commodity Y
        MKT_L   Supply-demand balance for primary factor L
        MKT_K   Supply-demand balance for primary factor K
        MKT_W   Supply-demand balance for aggregate demand

        I_CONS  Income definition for CONS;

*       Zero profit conditions:

PRF_X..  80*PLS**0.4 * PKS**0.6 =G= 120*PX;

PRF_Y..  80*PLS**0.6 * PKS**0.4 =G= 120*PY;

PRF_TL.. 80*PL*(1+TXL) =G= 80*PLS;

PRF_TK.. 80*PK*(1+TXK) =G= 80*PKS;

PRF_W.. 340*(PX)**(12/34) * (PY)**(12/34) * PL**(10/34)
            =G= 340 * PW;
```

```
*       Market clearing conditions:

MKT_X..  120*X =G= 340*W*PW * (12/34)/PX;

MKT_Y..  120*Y =G= 340*W*PW * (12/34)/PY;

MKT_W..  340*W =G= CONS / PW;

MKT_L..  180  =G=  80*TL + 340*W*(10/34)*(PW/PL);

MKT_K..  80   =G=  80*TK;

MKT_TL.. 80*TL =G= 48*X*PX/PLS + 72*Y*PY/PLS;

MKT_TK.. 80*TK =G= 72*Y*PY/PKS + 48*X*PX/PKS;

*       Income constraints:

I_CONS.. CONS =E= 180*PL + 80*PK + 80*TL*TXL*PL + 80*TK*TXK*PK;

MODEL INCOMETAX /PRF_X.X, PRF_Y.Y, PRF_TK.TK,PRF_TL.TL,
                 PRF_W.W, MKT_X.PX, MKT_Y.PY, MKT_L.PL,
                 MKT_TK.PKS, MKT_TL.PLS,
                 MKT_K.PK, MKT_W.PW, I_CONS.CONS /;

X.L     =1;
```

```
Y.L     =1;
TK.L    =1;
TL.L    =1;
W.L     =1;

PL.L    =1;
PX.L    =1;
PY.L    =1;
PLS.L   =1.5;
PKS.L   =1.5;
PK.L    =1;
PW.FX   =1;
CONS.L  =340;

TXL     =0.5;
TXK     =0.5;

INCOMETAX.ITERLIM = 0;
SOLVE INCOMETAX USING MCP;

*       Lets do some counter-factual with taxes shifted to the
*       factor which is in fixed supply:

INCOMETAX.ITERLIM = 1000;
SOLVE INCOMETAX USING MCP;
```

```
LOOP(S,

TXL = 0.60 - 0.10*ORD(S);
TXK = 0.40 + 0.10*ORD(S);

SOLVE INCOMETAX USING MCP;

WELFARE(S) = W.L;
LABSUP(S) = TL.L;
INCOME(S) = ((PX.L/1.5)*X.L + (PY.L/1.5)*Y.L)
           /((PX.L/1.5)**0.5*(PY.L/1.5)**0.5)/2;
CAPTAX(S) = TXK;
TAXREV(S) = (TXL*PL.L*TL.L*80 + TXK*PK.L*TK.L*80)
           /((PX.L/1.5)**0.5*(PY.L/1.5)**0.5);

);

DISPLAY WELFARE, LABSUP, INCOME, CAPTAX, TAXREV;

PARAMETER
 RESULTS(S, *);

RESULTS(S, "WELFARE") = WELFARE(S);
RESULTS(S, "LABSUP") = LABSUP(S);
RESULTS(S, "TAXREV") = TAXREV(S);
```

```
DISPLAY RESULTS;


TXL = 0;
TXK = 0;

SOLVE INCOMETAX USING MCP;
```

6.2b Equal yield tax reform

We set up a model in which we can do differential tax policy analysis holding the level of government revenue constant

This model introduces a fourth (and final) class of variables (in addition to activity levels, commodity prices and income levels).

The new entity is called an "auxiliary variable". In this model, we use an auxiliary variable to endogenously alter the tax rate in order to maintain an equal yield.

In the present case, we will hold the labor tax rate *exogenous*, but change its value, solving for the value of the *endogenous* capital tax that yields the same value of revenue as the original tax.

TXK now become a *variable*, not a *parameter*.

In the initial statements specifying the variables and the equations the model, we declare an extra variable TXK and an extra equation A_TXK (“A” for auxiliary)

Here is the constraint equation as it appears in the model

```
A_TXK..  TXL*PL*TL*80 + TXK*PK*TK*80
          =E= 80 *((PX**0.5 * PY**0.5);
```

The left-hand side is tax revenue from the two taxes, one an exogenous parameter (TXL) and the other an endogenous variable (TXK).

Each term is (tax rate) * (factor price) * (activity level for factor supply) * (the reference quantity supplied at an activity level equal to one).

The right-hand side of the constraint specifies the target revenue. The modeler has to think carefully about what is meant by "constant" revenue: that is, constant in terms of what?

Assume that the government wants the taxes to yield an amount equal to the cost of purchasing a "composite" unit of (sub) utility from X and Y. The cost is given from the consumer's expenditure function as

```
80 *((PX**0.5 * PY**0.5);
```

Of course, the government is not actually buying anything in this simple model, it is just redistributing the revenue back to the consumer.

But the modeler must specify what the revenue target is in real terms.

In our case, the initial value of TXK = 0.50, so we set this along with the values of PLS and PKS which are equal to 1.5 initially, along with the initial value of the parameter TXL (the latter is a parameter and so does not use the ‘.L’ syntax).

```
PX.L = 1.;
PY.L = 1.;
PLS.L = 1.5;
PKS.L = 1.5;
TXL   = 0.50;
TXK.L = 0.50;
```

After the replication check, we loop over values of TXL, and each solve statement finds the new value of TXK as one variable in the new general-equilibrium solution.

In each iteration, we store the values of key variables so that they can be presented together at the end of the listing file.

We include the difference between the effects of the reform on real commodity consumption (REALCONS) and true welfare (WELFARE), the latter accounting the value of leisure.

Note from the results in the present case, that measuring only the change in real commodity consumption significantly overstates the true welfare gain of the tax reform (which is tiny) because of the fall in leisure (increase in labor supply).

```
$TITLE  M6-2b.GMS: 2x2 Economy with income tax, endogenous tax rate
*       adds equal yield tax reform to model M6-2a

$ONTEXT
Illustrates equal yield tax reform to introduce auxiliary
 variable and constraint equaltion
Distorionary labor tax is lowered and capital tax raised
 endogenously (TXK is now a VARIABLE) to hold revenue constant

                  Production Sectors                Consumers

   Markets   |    X       Y        W      TL    TK       CONS
   ----------------------------------------------------------
        PX   |  120              -120
        PY   |          120      -120
        PW   |                    340                    -340
        PLS  |  -48     -72                120
        PKS  |  -72     -48                      120
        PL   |                   -100      -80            180
        PK   |                                   -80       80
        TAX  |                             -40   -40       80
   ----------------------------------------------------------
$OFFTEXT

SETS S /1*6/;
```

```
PARAMETERS
  TXL           Labor income tax rate
  WELFARE(S)    Welfare
  LABSUP(S)     Labor supply
  INCOME(S)     Money income = consumption of X and Y
  CAPTAX(S)     Endogenous capital tax for equal yield
  TAXREV(S)     Tax revenue in terms of purchasing power;


POSITIVE VARIABLES

   X        Activity level for sector X
   Y        Activity level for sector Y
   TL       Supply activity for L
   TK       Supply activity for K
   W        Activity level for sector W
   PX       Price index for commodity X
   PY       Price index for commodity Y
   PL       Price index for primary factor L net of tax
   PK       Price index for primary factor K net of tax
   PLS      Price index for primary factor L gross of tax
   PKS      Price index for primary factor K gross of tax
   PW       Price index for welfare (expenditure function)
   CONS     Income definition for CONS
   TXK      Endogenous capital tax from equal yield constraint;
```

```
EQUATIONS

   PRF_X   Zero profit for sector X
   PRF_Y   Zero profit for sector Y
   PRF_TL  Zero profit for sector TL
   PRF_TK  Zero profit for sector TK
   PRF_W   Zero profit for sector W

   MKT_X   Supply-demand balance for commodity X
   MKT_TK  Supply-demand balance for commodity TK
   MKT_TL  Supply-demand balance for commodity TL
   MKT_Y   Supply-demand balance for commodity Y
   MKT_L   Supply-demand balance for primary factor L
   MKT_K   Supply-demand balance for primary factor K
   MKT_W   Supply-demand balance for aggregate demand

   I_CONS  Income definition for CONS
   A_TXK   Auxiliary eq associated with equal yield constraint;

*       Zero profit conditions:

PRF_X..  80*PLS**0.4 * PKS**0.6 =G= 120*PX;

PRF_Y..  80*PLS**0.6 * PKS**0.4 =G= 120*PY;

PRF_TL.. 80*PL*(1+TXL) =G= 80*PLS;
```

```
PRF_TK.. 80*PK*(1+TXK) =G= 80*PKS;

PRF_W.. 340*(PX)**(12/34) * (PY)**(12/34) * PL**(10/34)
             =G= 340 * PW;

*       Market clearing conditions:

MKT_X..  120*X =G= 340*W*PW * (12/34)/PX;

MKT_Y..  120*Y =G= 340*W*PW * (12/34)/PY;

MKT_W..  340*W =G= CONS / PW;

MKT_L..  180  =G=  80*TL + 340*W*(10/34)*(PW/PL);

MKT_K..  80   =G=  80*TK;

MKT_TL.. 80*TL =G= 48*X*PX/PLS + 72*Y*PY/PLS;

MKT_TK.. 80*TK =G= 72*Y*PY/PKS + 48*X*PX/PKS;

*       Income constraints:

I_CONS.. CONS =E= 180*PL + 80*PK + 80*TL*TXL*PL + 80*TK*TXK*PK;
```

```
A_TXK..  TXL*PL*TL*80 + TXK*PK*TK*80
          =E= 80 *(PX**0.5 * PY**0.5);


MODEL ALGEBRAIC /PRF_X.X, PRF_Y.Y, PRF_TK.TK,PRF_TL.TL,
                 PRF_W.W, MKT_X.PX, MKT_Y.PY, MKT_L.PL,
                 MKT_TK.PKS, MKT_TL.PLS,
                 MKT_K.PK, MKT_W.PW, I_CONS.CONS, A_TXK.TXK /;

X.L     =1;
Y.L     =1;
TK.L    =1;
TL.L    =1;
W.L     =1;

PL.L    =1;
PX.L    =1;
PY.L    =1;
PLS.L   =1.5;
PKS.L   =1.5;
PK.L    =1;
PW.FX   =1;
CONS.L  =340;

TXL     =0.5;
TXK.L   =0.5;
```

```
ALGEBRAIC.ITERLIM = 0;
SOLVE ALGEBRAIC USING MCP;

*     Lets do some counter-factual with taxes shifted to the
*     factor which is in fixed supply:

ALGEBRAIC.ITERLIM = 1000;
SOLVE ALGEBRAIC USING MCP;

LOOP(S,

TXL = 0.60 - 0.10*ORD(S);

SOLVE ALGEBRAIC USING MCP;

WELFARE(S) = W.L;
LABSUP(S) = TL.L;
INCOME(S) = (PX.L*X.L + PY.L*Y.L)
           /(PX.L**0.5*PY.L**0.5)/2;
CAPTAX(S) = TXK.L;

TAXREV(S) = (TXL*PL.L*TL.L*80 + TXK.L*PK.L*TK.L*80)
           /(PX.L**0.5*PY.L**0.5));

DISPLAY WELFARE, LABSUP, INCOME, CAPTAX, TAXREV;
```

```
TXL = 0;
TXK.FX = 0;

SOLVE ALGEBRAIC USING MCP;
```

6.3 Public consumption goods

The assumption of lump-sum redistribution is a convenient trick which simplifies tax policy analysis.

In practice, governments often use money to purchase things which private markets cannot provide..

In this model, we first explicitly introduce government as an agent or “consumer” (GOVT).

The tax revenue collected in the economy is assigned to the government.

The government spends this on purchasing a good called G (price PG), produced from capital and labor like goods X and Y.

Sectors	Production					Consumers		
Markets	X	Y	G	W1	W2	CONS1	CONS2	GOVT
PX	100			-50	-50			
PY		100		-50	-50			
PG			50					-50
PL	-80	-80	-40			100	100	
TAX	-20	-20	-10					50
PW1				125		-125		
PW2					125		-125	
PG1				-25		25		
PG2					-25		25	

The government is the only agent demanding PG in the model and taxes are the government’s only source of income.

Each consumer receives the full benefit of the public good: a public good is non-rivaled.

And a consumer cannot be charged for the good, nor can a consumer sell it to other consumers: a public good is non-excludable

The way that we co this in GAMS is to have the government buy the good from tax revenue, then the full amount of the good is transferred or endowed to each consumer.

This is done via an auxiliary variable and a constraint equation.

```
LGP =E= G;
```

where G is production of the good and LGP is each consumer's endowment of the good, viewed as exogenous.

Since each consumer's endowment of the good is fixed and equal,, consumers with different incomes or preferences will in general have different demand prices for the good.

These are often referred to as "willingness to pay".

PG1 and PG2 are the demand prices or willingness to pay by consumers 1 and 2 respectively.

These are in effect separate or "personalized" goods: one consumer cannot sell his/her good to the other consumer.

We can capture the non-excludability and non-rivaled properties by thinking of two separate "markets": each consumer "demands" the good they are endowment with which, because the quantity is fixed, allows us to solve for each consumer's demand price separately.

Here are the relevant, key equations, where again, LGP is viewed as exogenously by each consumer.

```
MKT_G1.. 50*LGP =G= 50 * W1 * PW1*0.5/PG1;

MKT_G2.. 50*LGP =G= 50 * W2 * PW2*0.5/PG2;

I_CONS1.. CONS1 =E= 50*PL + 50*PK + 50*LGP*PG1;

I_CONS2.. CONS2 =E= 50*PL + 50*PK + 50*LGP*PG2;
```

Optimality: the optimal provision of a public good occurs when the marginal cost of providing the good equals the <u>sum</u> of the consumers’ willingness to pay (since equal consumer gets the full benefit of an additional unit: non-rivaled).

In the data, we have assumed that the valuations of the public goods at a price PG1 = PG2 = 0.5 and th marginal cost of provision is PG = 1.

Thus the initial data represent an optimal initial provision of the public good. PG = PG1 + PG2.

Note that this is an <u>assumption</u>. We do not actually observed the demand prices (willingness to pay) in any real data.

(A big task in environmental economies is to estimate willingness to pay for various goods; e.g., parks and open space.)

```
$TITLE  M6-3: Economy with two households and a public good

$ONTEXT
How do we model a public good that is non-excludable and non-rivaled?

             Production Sectors          Consumers

Markets|    X    Y    G   W1   W2       CONS1   CONS2   GOVT
--------------------------------------------------------------
  PX   |  100            -50  -50
  PY   |       100       -50  -50
  PG   |            50                                    -50
  PL   |  -80  -80  -40                   100     100

  TAX  |  -20  -20  -10                                    50

  PW1  |                 125             -125
  PW2  |                      125                -125
  PG1  |                 -25                25
  PG2  |                      -25                  25
--------------------------------------------------------------
$OFFTEXT

PARAMETER
    TAX     Value-added tax rate;
```

```
NONNEGATIVE VARIABLES
    X        Activity level for sector X
    Y        Activity level for sector Y
    W1       Activity level for sector W1
    W2       Activity level for sector W2
    G        Activity level for government sector

    PX       Price index for commodity X
    PY       Price index for commodity Y
    PL       Price index for primary factor L
    PW1      Price index for welfare 1(expenditure function)
    PW2      Price index for welfare 2(expenditure function)
    PG1      Private valuation of the public good (consumer 1)
    PG2      Private valuation of the public good (consumer 2)
    PG       Price of (cost of producing) the public good

    GOVT     Budget restriction for government
    CONS1    Income definition for CONS1
    CONS2    Income definition for CONS2

    LGP      Endowment of public good received by each consumer;

EQUATIONS
    PRF_X    Zero profit for sector X
    PRF_Y    Zero profit for sector Y
    PRF_W1   Zero profit for sector W1
```

```
    PRF_W2  Zero profit for sector W2
    PRF_G   Zero profit in government sector

    MKT_X   Supply-demand balance for commodity X
    MKT_Y   Supply-demand balance for commodity Y
    MKT_L   Supply-demand balance for primary factor L
    MKT_W1  Supply-demand balance for consumer 1
    MKT_W2  Supply-demand balance for consumer 2
    MKT_G1  Private valuation of the public good (consumer 1)
    MKT_G2  Private valuation of the public good (consumer 2)
    MKT_G   Supply-demand balance for commodity G

    I_G     Budget restriction for government
    I_CONS1 Income definition for CONS1
    I_CONS2 Income definition for CONS2

    A_LGP   Auxiliary for government provision;

*       Zero profit conditions:

PRF_X..  80*PL * (1+TAX) =G= 100*PX;

PRF_Y..  80*PL * (1+TAX) =G= 100*PY;

PRF_G..  40*PL * (1+TAX) =G= 50*PG;
```

```
PRF_W1.. 125*PX**(50/125) * PY**(50/125) * (PG1/0.5)**(25/125)
             =G= 125*PW1;

PRF_W2.. 125*PX**(50/125) * PY**(50/125) * (PG2/0.5)**(25/125)
             =G= 125*PW2;

*       Market clearing conditions:

MKT_X..  100*X =G= 50*W1*PW1/PX + 50*W2*PW2/PX ;

MKT_Y..  100*Y =G= 50*W1*PW1/PY + 50*W2*PW2/PY;

MKT_L..  200 =G= (80*X + 80*Y + 40*G);

MKT_W1.. 125*W1 =G= CONS1 / PW1;

MKT_W2.. 125*W2 =G= CONS2 / PW2;

MKT_G..  50*G =G= GOVT/ PG;

MKT_G1.. 50*LGP =G= 25 * W1 * PW1/PG1;

MKT_G2.. 50*LGP =G= 25 * W2 * PW2/PG2;

*       Income constraints:
```

```
I_G..      GOVT =G= PL*(80*X + 80*Y + 40*G )*TAX;

I_CONS1.. CONS1 =E= 100*PL + 50*LGP*PG1;

I_CONS2.. CONS2 =E= 100*PL + 50*LGP*PG2;

*       Auxiliary constraints:

A_LGP..    LGP =E= G;

MODEL PUBGOOD /PRF_X.X, PRF_Y.Y, PRF_W1.W1, PRF_W2.W2, PRF_G.G,
                 MKT_X.PX, MKT_Y.PY, MKT_L.PL,
                 MKT_W1.PW1, MKT_W2.PW2,
                 MKT_G.PG, MKT_G1.PG1, MKT_G2.PG2,
                 I_G.GOVT, I_CONS1.CONS1, I_CONS2.CONS2,
                 A_LGP.LGP /;

X.L     =1;
Y.L     =1;
W1.L    =1;
W2.L    =1;
G.L     =1;

PL.FX   =1;
PX.L    =1;
PY.L    =1;
```

```
PG.L    =1;
PW1.L   =1;
PW2.L   =1;
PG1.L   =0.5;
PG2.L   =0.5;

CONS1.L =125;
CONS2.L =125;
GOVT.L  =50;

LGP.L   =1;

TAX     =0.25;

PUBGOOD.ITERLIM = 0;
SOLVE PUBGOOD USING MCP;
PUBGOOD.ITERLIM = 2000;
SOLVE PUBGOOD USING MCP;

*    The following counterfactuals check that the original
*    benchmark is indeed an optimum by
*    raising/lowering the tax

TAX = 0.10;

SOLVE PUBGOOD USING MCP;
```

```
TAX = 0.40;

SOLVE PUBGOOD USING MCP;
```

6.4 Optimal provision using a Samuelson rule

This model is exactly the same as the previous one, except that the tax used to finance the public good is endogenous.

Instead of TAX being a parameter, it is now an auxiliary variable. Its value is set by the constraint equation:

```
PG =E= PG1 + PG2;
```

Since each consumer gets the full amount of the public good (the good is “non-rivaled”), the marginal benefit of another unit of the good is the sum of the demand prices for all the consumers.

Efficiency is achieved when this sum of benefits is equal to the marginal cost of producing another unit.

Note that the auxiliary variable itself need not appear in the constraint equation associated with it. The solution algorithm will adjust TAX in order to satisfy this condition.

Caveat: the Samuelson rule is valid only if the tax needed to pay for the public good can be raised in a non-distortionary way.

If distortionary taxes must be used, the sum of marginal benefits must be weighed against the marginal cost of production plus the marginal burden of taxation.

When we run this model, we will get back a value of TAX = 0.25, because we calibrated the preferences assuming that the initial data was optimal.

As a counterfactual experiment, we change one consumer’s valuation of the public good, by changing the share parameters in consumer 1's utility function.

Share parameters in the benchmark are set and declared as:

```
PARAMETERS
 SHX1, SHY1, SHG1  shares of X Y and G in 1's utility
 SHX2, SHY2, SHG2  shares of X Y and G in 2's utility;

SHG1 = 0.2;
SHX1 = 0.5 - SHG1/2;
SHY1 = 0.5 - SHG1/2;
SHG2 = 0.2;
SHX2 = 0.5 - SHG2/2;
SHY2 = 0.5 - SHG2/2;
```

Counterfactual experiment sets:

```
SHG1 = 0.3;
SHX1 = 0.5 - SHG1/2;
SHY1 = 0.5 - SHG1/2;
```

Note that, although the higher tax is efficient according to the Samuelson rule, it nevertheless results in a redistribution of welfare from the low valuation consumer to the high valuation consumer.

```
$TITLE  M6-4.GMS: Economy with two consumers, public good,
*        optimal provision with an endogenous tax rate, Samuelson rule

$ONTEXT
 Samuelson rule for optimal provision, PG = PG1 + PG2
  introduces an auxiliary variable and constraint equation
 Here is the tax rate is a VARIABLE, set optimally
 Generalizes M6-3.gms: two consumers with different preferences

             Production Sectors          Consumers

Markets|    X    Y    G   W1   W2       CONS1   CONS2   GOVT
--------------------------------------------------------------
  PX   |  100            -50  -50
  PY   |       100       -50  -50
  PG   |             50                                  -50
  PL   |  -80  -80  -40                   100     100

  TAX  |  -20  -20  -10                                   50

  PW1  |                  125            -125
  PW2  |                       125               -125
  PG1  |                  -25              25
  PG2  |                       -25                 25
--------------------------------------------------------------
$OFFTEXT
```

```
PARAMETERS
 SHX1, SHY1, SHG1  shares of X Y and G in consumer 1's utility
 SHX2, SHY2, SHG2  shares of X Y and G in consumer 2's utility;

SHG1 = 0.2;
SHX1 = 0.5 - SHG1/2;
SHY1 = 0.5 - SHG1/2;
SHG2 = 0.2;
SHX2 = 0.5 - SHG2/2;
SHY2 = 0.5 - SHG2/2;

POSITIVE VARIABLES
   X       Activity level for sector X,
   Y       Activity level for sector Y,
   W1      Activity level for sector W1,
   W2      Activity level for sector W2,
   G       Activity level for government sector,

   PX      Price index for commodity X,
   PY      Price index for commodity Y,
   PL      Price index for primary factor L,
   PW1     Price index for welfare 1(expenditure function),
   PW2     Price index for welfare 2(expenditure function),
   PG1     Private valuation of the public good (consumer 1),
   PG2     Private valuation of the public good (consumer 2),
```

```
   PG       Price (marginal cost) of the public good

   GOVT     Budget restriction for government,
   CONS1    Income definition for CONS1,
   CONS2    Income definition for CONS2,

   LGP      Level of government provision
   TAX      Uniform value-added tax rate;

EQUATIONS
   PRF_X    Zero profit for sector X
   PRF_Y    Zero profit for sector Y
   PRF_W1   Zero profit for sector W1
   PRF_W2   Zero profit for sector W2
   PRF_G    Zero profit in government sector

   MKT_X    Supply-demand balance for commodity X
   MKT_Y    Supply-demand balance for commodity Y
   MKT_L    Supply-demand balance for primary factor L
   MKT_W1   Supply-demand balance for consumer 1
   MKT_W2   Supply-demand balance for consumer 2
   MKT_G1   Private valuation of the public good (consumer 1)
   MKT_G2   Private valuation of the public good (consumer 2)
   MKT_G    Supply-demand balance for commodity G

   I_G      Budget restriction for government
```

```
   I_CONS1 Income definition for CONS1
   I_CONS2 Income definition for CONS2

   A_LGP   Auxiliary for government provision
   A_TAX   Auxiliary for government provision;

*       Zero profit conditions:

PRF_X..  80*PL * (1+TAX) =G= 100*PX;

PRF_Y..  80*PL * (1+TAX) =G= 100*PY;

PRF_W1.. 125*PX**(SHX1) * PY**(SHY1) * (PG1/0.5)**(SHG1)
             =E= 125*PW1;

PRF_W2.. 125*PX**(SHX2) * PY**(SHY2) * (PG2/0.5)**(SHG2)
             =E= 125*PW2;

PRF_G..  40*PL * (1+TAX) =G= 50*PG;

*       Market clearing conditions:

MKT_X..  100*X =G= 125*SHX1*W1*PW1/PX + 125*SHX2*W2*PW2/PX ;

MKT_Y..  100*Y =G= 125*SHY1*W1*PW1/PY + 125*SHY2*W2*PW2/PY;
```

```
MKT_W1.. 125*W1 =G= CONS1/PW1;

MKT_W2.. 125*W2 =G= CONS2/PW2;

MKT_L..  200 =G= (80*X + 80*Y + 40*G);

MKT_G1.. 50 * LGP =G= 125*SHG1 * W1 * PW1/PG1;

MKT_G2.. 50 * LGP =G= 125*SHG2 * W2 * PW2/PG2;

MKT_G..  50*G =G= GOVT/ PG;

*       Income constraints:

I_G..      GOVT =G= PL*(80*X + 80*Y + 40*G )*TAX;

I_CONS1.. CONS1 =E= 100*PL + 50*LGP*PG1;

I_CONS2.. CONS2 =E= 100*PL + 50*LGP*PG2;

*       Auxiliary constraints:

A_LGP..  LGP =E= G;

A_TAX..  PG =E= PG1 + PG2;
```

```
MODEL PUBGOOD2 /PRF_X.X, PRF_Y.Y, PRF_W1.W1, PRF_W2.W2,
                 PRF_G.G,
                 MKT_X.PX, MKT_Y.PY, MKT_L.PL,
                 MKT_W1.PW1, MKT_W2.PW2,
                 MKT_G.PG, MKT_G1.PG1, MKT_G2.PG2,
                 I_G.GOVT, I_CONS1.CONS1, I_CONS2.CONS2,
                 A_LGP.LGP, A_TAX.TAX /;

X.L     =1;
Y.L     =1;
W1.L    =1;
W2.L    =1;
G.L     =1;

PL.FX   =1;
PX.L    =1;
PY.L    =1;

PG.L    =1;
PW1.L   =1;
PW2.L   =1;
PG1.L   =0.5;
PG2.L   =0.5;

CONS1.L =125;
```

```
CONS2.L =125;
GOVT.L  =50;

LGP.L   =1;

TAX.L   =0.25;

PUBGOOD2.ITERLIM = 0;
SOLVE PUBGOOD2 USING MCP;
PUBGOOD2.ITERLIM = 2000;
SOLVE PUBGOOD2 USING MCP;

*   Change consumer 1's preferences, higher preference for the
*   public good, which now has a Cobb-Douglas share of 0.3

SHG1 = 0.3;
SHX1 = 0.5 - SHG1/2;
SHY1 = 0.5 - SHG1/2;

*PUBGOOD2.ITERLIM = 0;
SOLVE PUBGOOD2 USING MCP;


TAX.FX = 0.25;

SOLVE PUBGOOD2 USING MCP;
```

6.5 Public intermediate (infrastructure) good with optimal provision

Suppose that output in the X sector is given by

$$X = \alpha L,$$

where L is a private input and α is a parameter which is increasing in the level of a government-provided infra-structure good.

Individual firms view α as exogenous.

Producing one unit of X then requires $1/\alpha$ units of L. The unit cost function for X is then $p_l/\alpha = p_x$.

The public good G is produced from labor only (the only factor of production), and is financed by an equal tax on all goods (including the public good).

The equation for alpha is given by

```
ALPHA =E= 1 + INFPROD*G;
```

where INFPROD is a parameter giving productivity of G in X.

The marginal product of G in producing X (L held constant), is then

$$\frac{\partial X}{\partial G} = INFPROD * L$$ where L is the labor used in X

Referring back to the production function, we can replace L with

$$\frac{\partial X}{\partial G} = INFPROD * (X/\alpha)$$

Now multiply this by PX to get the value of the marginal product of G in X. This should then be set equal to the price (marginal cost) of a unit of G, PG.

$$p_g = p_x INFPROD * (X/\alpha)$$

This will be an auxiliary equation that sets a non-distortionary (endogenous) income tax rate TX to its optimal value.

```
$TITLE  M6-5.GMS: Public intermediate good with optimal provision
* technique for modeling infrastructure for example

$ONTEXT
             Production Sectors          Consumers

Markets|    X     Y     G    W1        CONS1     GOVT
--------------------------------------------------------------
  PX   |  100              -100
  PY   |       100         -100
  PG   |              50                          -50
  PL   |  -80  -80   -40                  200
  TAX  |  -20  -20   -10                           50
  PW   |                    200          -200
--------------------------------------------------------------
  X = ALPHA*L  ALPHA = F(G)  ALPHA viewed as exogenous by firms

$OFFTEXT

PARAMETERS
 SHX, SHY  shares of X and Y in consumer's utility
 INFPROD   productivity parameter of the public good in X output
 WELF;

SHX = 0.5;
SHY = 0.5;
```

```
INFPROD = 0;

POSITIVE VARIABLES
   X        Activity level for sector X
   Y        Activity level for sector Y
   W        Activity level for sector W
   G        Activity level for government sector

   PX       Price index for commodity X
   PY       Price index for commodity Y
   PG       Private valuation of the public good
   PL       Price index for primary factor L
   PW       Price index for welfare 1(expenditure function)

   GOVT     Budget restriction for government
   CONS     Income definition for CONS

   TAX      Uniform value-added tax rate
   ALPHA    Public intermediary good multiplier on productivity;

EQUATIONS
   PRF_X    Zero profit for sector X
   PRF_Y    Zero profit for sector Y
   PRF_W    Zero profit for sector W1
   PRF_G    Zero profit in government sector
```

```
   MKT_X   Supply-demand balance for commodity X
   MKT_Y   Supply-demand balance for commodity Y
   MKT_G   Supply-demand balance for commodity G
   MKT_L   Supply-demand balance for primary factor L
   MKT_W   Supply-demand balance for consumer 1

   I_G     Budget restriction for government
   I_CONS  Income definition for CONS

   A_TAX   Auxiliary for government provision
   INFRA   Auxiliary for public intermediate good calculation;

*       Zero profit conditions:

PRF_X..  80*PL * (1+TAX)/ALPHA =G= 100*PX;

PRF_Y..  80*PL * (1+TAX) =G= 100*PY;

PRF_W .. 200*PX**(SHX) * PY**(SHY) =E= 200*PW;

PRF_G..  40*PL * (1+TAX) =G= 100*PG;

*       Market clearing conditions:

MKT_X..  100*X =G= 200*SHX*W*PW/PX;
```

```
MKT_Y..  100*Y =G= 200*SHY*W*PW/PY;

MKT_G..  100*G =G= GOVT/ PG;

MKT_L..  200 =G= (80*X/ALPHA + 80*Y + 40*G);

MKT_W..  200*W =G= CONS/PW;

*       Income constraints:

I_G..    GOVT =G= PL*(80*X/ALPHA + 80*Y + 40*G )*TAX;

I_CONS.. CONS =E= 200*PL;

*       Auxiliary constraints:

A_TAX..  PG =E= PX*INFPROD*(X/ALPHA);

INFRA..  ALPHA =E= 1 + INFPROD*G;


MODEL PUBINT    /PRF_X.X, PRF_Y.Y, PRF_W.W, PRF_G.G,
                 MKT_X.PX, MKT_Y.PY, MKT_L.PL, MKT_W.PW, MKT_G.PG,
                 I_G.GOVT, I_CONS.CONS,
                 A_TAX.TAX, INFRA.ALPHA /;
```

```
X.L      =1;
Y.L      =1;
W.L      =1;
G.L      =1;
PL.FX    =1;
PX.L     =1;
PY.L     =1;
PG.L     =0.5;
PW.L     =1;
CONS.L   =200;
GOVT.L   =50;
ALPHA.L = 1;
TAX.L    = .25;

PUBINT.ITERLIM = 0;
SOLVE PUBINT USING MCP;

* with INFPROD = 0 initially, the optimal tax should be zero

PUBINT.ITERLIM = 2000;
SOLVE PUBINT USING MCP;

* now set INFPROD = 2, optimal tax and provision should be positive

INFPROD = 2;
TAX.L = 0.25; G.L = 1;
```

```
SOLVE PUBINT USING MCP;

WELF = W.L*100;
DISPLAY WELF;


* now let's check by "brute force" whether the answer is right
* loop over fixed values of TAX

SETS I /I1*I15/;

PARAMETERS
 WELFARE(I)
 TAXRATE(I);

LOOP(I,
TAX.FX = 0.29 + 0.01*ORD(I);

SOLVE PUBINT USING MCP;

WELFARE(I) = 100*W.L;
TAXRATE(I) = TAX.L;

);

DISPLAY TAXRATE, WELFARE;
```

6.6a Pollution from production affects utility

This model is: two goods, one factor, one consumer

Pollution is generated by the production of X, pollution reduces utility

Pollution is modeled as a reduction in the endowment of CLEAN AIR
Initial endowment of clear air is 200, with 100 reduced by X pollution and 100 entering utility. PCA = price of clean air.

	Production Sectors			Consumers
Markets	X	Y	W	CONS
PX	100		-100	
PY		100	-100	
PW			300	-300
PL	-100	-100		200
PCA			-100	(200 - 100)

As in the case of a public good, this public “bad” must be modeled as non-rivaled and non-excludable. (Non-rivaled is trivial here since there is only one consumer.)

The utility function gives 1/3 equal weights to X, Y, and CA. Expenditure function is given by:

```
PRF_W.. 200*(PX**(1/3) * PY**(1/3) * PCA**(1/3)) =G= 200*PW;
```

Shepard’s lemma then gives a demand for clean air: as in the public good case, consumer’s cannot actually chose;

Rather, this gives a demand price for the given amount of clean air. This “willingness to pay” is part of the solution to the model.

The supply of clean air is given as the endowment 200, minus that which is “stolen” by pollution from the production of X: 100*POL.

```
MKT_CA.. 200-100*POL =G= 100 * W * PW / PCA;
```

Consumer income will be defined as inclusive of the value of clean air, similar to our treatment of leisure. TX is a pollution tax on X.

```
I_CONS..  CONS =E= 200*PL + (200-100*POL)*PCA + TX*100*X*PL;
```

Pollution is proportional to the production of X. POLINT is a parameter for pollution intensity of X produciton.

```
PPOL..   100*POL =G= POLINT*100*X;
```

```
$TITLE:  M6-6a.GMS: Modelling pollution as reducing the endowment
*        of an environment public good

$ONTEXT
This model is a closed economy: two goods and one factor, one consumer
Pollution is generated by the production of X, pollution reduces utility
Pollution is modeled as a reduction in the endowment of CLEAN AIR
Initial endowment of clear air is 200, with 100 reduced by X pollution
 and 100 entering utility.

                  Production Sectors          Consumers
   Markets   |    X        Y         W    |       CONS
   ------------------------------------------------------
        PX   |  100                -100   |
        PY   |          100        -100   |
        PW   |                      300   |       -300
        PL   | -100    -100               |        200
   ______________________________________________________
        PCA                        -100              (200 - 100)
   ------------------------------------------------------
$OFFTEXT

PARAMETERS

 TX          ad-valorem tax rate for X sector inputs
 POLINT      polution intensity multiplier;
```

```
TX = 0;
POLINT = 1;

POSITIVE VARIABLES

 X        activity level for X production
 Y        activity level for Y production
 W        activity level for the "production" of welfare from X Y

 PX       price of good X
 PY       price of good Y
 PCA      price of clean air
 PW       price of a unit of welfare (real consumer-price index)
 PL       price of labor

 CONS     income of the representative consumer
 POL      pollution;

EQUATIONS

 PRF_X    zero profit for sector X
 PRF_Y    zero profit for sector Y
 PRF_W    zero profit for sector W (Hicksian welfare index)

 MKT_X    supply-demand balance for commodity X
```

```
 MKT_Y    supply-demand balance for commodity Y
 MKT_CA   market for clean air (determines shadow value PCA)
 MKT_L    supply-demand balance for primary factor L
 MKT_W    supply-demand balance for aggregate demand

 I_CONS   income definition for CONS
 PPOL     pollution caused by production - consumption of X;

*       Zero profit inequalities

PRF_X..   100*PL*(1+TX) =G= 100*PX;

PRF_Y..   100*PL =G= 100*PY;

PRF_W..   300*(PX**(1/3) * PY**(1/3) * PCA**(1/3)) =G= 300*PW;

*       Market clearance inequalities

MKT_X..   100*X        =G= 100 * W * PW / PX;

MKT_Y..   100*Y        =G= 100 * W * PW / PY;

MKT_CA..  200-100*POL =G= 100 * W * PW / PCA;

MKT_W..   300*W =E= CONS / PW;
```

```
MKT_L..   200 =G= 100*X + 100*Y;

*       Income balance equations (don't forget tax revenue)

I_CONS..  CONS =E= 200*PL + (200-100*POL)*PCA + TX*100*X*PL;

PPOL..    100*POL =G= POLINT*100*X;

MODEL POLLUTE /PRF_X.X, PRF_Y.Y, PRF_W.W,
                 MKT_X.PX, MKT_Y.PY, MKT_CA.PCA, MKT_L.PL,
                 MKT_W.PW,I_CONS.CONS, PPOL.POL /;


*       Chose a numeraire: real consumer price index

PW.FX = 1;

*       Set initial values of variables:

X.L=1; Y.L=1; W.L=1;
PX.L=1; PY.L=1; PL.L=1; POL.L = 1; PCA.L = 1;
CONS.L=300;

POLLUTE.ITERLIM = 0;
SOLVE POLLUTE USING MCP;
```

```
POLLUTE.ITERLIM = 1000;
SOLVE POLLUTE USING MCP;

*   counterfactual 1: 50% tax

TX = 0.5;
SOLVE POLLUTE USING MCP;

TX = 0.75;
SOLVE POLLUTE USING MCP;
```

6.6b Uses MPEC to solve for the optimal pollution tax

Now we make TX a variable rather than a tax. Second, we introduce another (unbounded) variable WELFARE (to be optimized). WELFARE just equals W from M6-6a.

This is an MPEC (optimization problem subject to equilibrium constraints). There is no need for an added equation for the added variable TX. The solver will find its optimal value.

The model has one unmatched equation (WELFARE), with the constraint set the same general-equilibrium model of M6-6a.

TX is not matched to an equation.

```
$TITLE: M6-6b.GMS: Pollution modelled as an MPEC to solve for optimal TX

$ONTEXT
Follows from M6-5a: two goods and one factor, one consumer
Pollution is generated by the production of X, pollution reduces utility
Pollution is modeled as a reduction in the endowment of CLEAN AIR
Initial endowment of clear air is 200, with 100 reduced by X pollution
 and 100 entering utility.
Solves for the welfare maximizing level of the pollution tax

                  Production Sectors           Consumers
   Markets   |    X       Y        W    |        CONS
   ------------------------------------------------------
        PX   |  100              -100   |
        PY   |          100      -100   |
        PW   |                    300   |        -300
        PL   | -100    -100             |         200
   ______________________________________________________
        PCA                      -100              (200 - 100)
   ------------------------------------------------------
$OFFTEXT

PARAMETERS

 POLINT     polution intensity multiplier;
```

```
POLINT = 1;

VARIABLES
 WELFARE   welfare
 TX        pollution tax on X;

POSITIVE VARIABLES

 X       activity level for X production
 Y       activity level for Y production
 W       activity level for the "production" of welfare from X Y

 PX      price of good X
 PY      price of good Y
 PCA     price of clean air
 PW      price of a unit of welfare (real consumer-price index)
 PL      price of labor

 CONS    income of the representative consumer
 POL     pollution;

EQUATIONS

 OBJ     Objective function: maximize welfare
 PRF_X   zero profit for sector X
 PRF_Y   zero profit for sector Y
```

```
 PRF_W   zero profit for sector W (Hicksian welfare index)

 MKT_X   supply-demand balance for commodity X
 MKT_Y   supply-demand balance for commodity Y
 MKT_CA  market for clean air (determines shadow value PCA)
 MKT_L   supply-demand balance for primary factor L
 MKT_W   supply-demand balance for aggregate demand

 I_CONS  income definition for CONS
 PPOL    pollution caused by production - consumption of X;

*       Zero profit inequalities

OBJ..     WELFARE =E= W;

PRF_X..   100*PL*(1+TX) =G= 100*PX;

PRF_Y..   100*PL =G= 100*PY;

PRF_W..   200*(PX**(1/3) * PY**(1/3) * PCA**(1/3)) =G= 200*PW;

*       Market clearance inequalities

MKT_X..   100*X       =G= 100 * W * PW / PX;

MKT_Y..   100*Y       =G= 100 * W * PW / PY;
```

```
MKT_CA..  200-100*POL =G= 100 * W * PW / PCA;

MKT_W..   300*W =E= CONS / PW;

MKT_L..   200 =G= 100*X + 100*Y;

*       Income balance equations (don't forget tax revenue)

I_CONS..  CONS =E= 200*PL + (200-100*POL)*PCA + TX*100*X*PL;

PPOL..    100*POL =G= POLINT*100*X;

MODEL POLLUTE /  OBJ, PRF_X.X, PRF_Y.Y, PRF_W.W,
                 MKT_X.PX, MKT_Y.PY, MKT_CA.PCA, MKT_L.PL,
                 MKT_W.PW,I_CONS.CONS, PPOL.POL /;


*       Chose a numeraire: real consumer price index

PW.FX = 1;

*       Set initial values of variables:

X.L=1; Y.L=1; W.L=1;
PX.L=1; PY.L=1; PL.L=1; POL.L = 1; PCA.L = 1;
```

```
CONS.L=300; WELFARE.L = 1;

OPTION MPEC = nlpec;
POLLUTE.ITERLIM = 0;
SOLVE POLLUTE USING MPEC MAXIMIZING WELFARE;

TX.L = 0.3;
WELFARE.L = 1.2;

POLLUTE.ITERLIM = 1000;
SOLVE POLLUTE USING MPEC MAXMIZING WELFARE;

* make pollution worse

POLINT = 1.5;

SOLVE POLLUTE USING MPEC MAXMIZING WELFARE;
```

6.6c Optimal tax set by a Pigouvian tax formula

Another way to find the optimal tax is to use a Pigouvian tax rule, which states that the price of the polluting good must equal its full cost.

In our case, this is the price of the privates inputs (labor) needed to produce one unit of X plus the marginal damages of pollution from one more unit of X.

So now we add an equation (and drop WELFARE) which is matched to the variable TX. This is given by:

```
ATX..           PX =E= PL + PCA*POLINT;
```

or noting that PX = PL*(1 + TX), the equation can be written as:

```
ATX..           TX =E= PCA*POLINT/PL;
```

```
$TITLE  M6-6c.GMS: Pollution tax set optimally via a
*  "first-order condition"
*  TX is set by an equation equation the price of X to it's full cost:
*  PX = PL + PCA

$ONTEXT
This model is a closed economy: two goods and one factor, one consumer
Pollution is generated by the production of X, pollution reduces utility
Pollution is modeled as a reduction in the endowment of CLEAN AIR
Initial endowment of clear air is 200, with 100 reduced by X pollution
 and 100 entering utility.

                  Production Sectors          Consumers
   Markets   |    X       Y        W    |       CONS
   ------------------------------------------------------
        PX   |  100              -100   |
        PY   |          100      -100   |
        PW   |                    300   |       -300
        PL   | -100    -100             |        200
   ______________________________________________________
        PCA                      -100           (200 - 100)
   ------------------------------------------------------

$OFFTEXT
```

```
PARAMETERS

 POLINT     polution intensity multiplier
 WELOPT     welfare under the optimal tax
 TAXOPT     value of the optimal tax;

POLINT = 1;


NONNEGATIVE VARIABLES

 X       activity level for X production
 Y       activity level for Y production
 W       activity level for the "production" of welfare from X Y

 PX      price of good X
 PY      price of good Y
 PCA     price of clean air
 PW      price of a unit of welfare (real consumer-price index)
 PL      price of labor

 CONS    income of the representative consumer
 POL     pollution
 TX      pollution tax;
```

```
EQUATIONS

 PRF_X   zero profit for sector X
 PRF_Y   zero profit for sector Y
 PRF_W   zero profit for sector W (Hicksian welfare index)

 MKT_X   supply-demand balance for commodity X
 MKT_Y   supply-demand balance for commodity Y
 MKT_CA  market for clean air (determines shadow value PCA)
 MKT_L   supply-demand balance for primary factor L
 MKT_W   supply-demand balance for aggregate demand

 I_CONS  income definition for CONS
 PPOL    pollution caused by production - consumption of X
 ATX     sets pollution tax optimally;

*       Zero profit inequalities

PRF_X..      100*PL*(1+TX) =G= 100*PX;

PRF_Y..      100*PL =G= 100*PY;

PRF_W..      200*(PX**(1/3) * PY**(1/3) * PCA**(1/3)) =G= 200*PW;

*       Market clearance inequalities
```

```
MKT_X..       100*X    =G= 100 * W * PW / PX;

MKT_Y..       100*Y    =G= 100 * W * PW / PY;

MKT_CA..      200-100*POL =G= 100 * W * PW / PCA;

MKT_W..       300*W =E= CONS / PW;

MKT_L..       200 =G= 100*X + 100*Y;

*       Income balance equations (don't forget tax revenue)

I_CONS..      CONS =E= 200*PL + (200-100*POL)*PCA + TX*100*X*PL;

PPOL..        100*POL =G= POLINT*100*X;

ATX..         PX =E= PL + PCA*POLINT;

* or since PX = PL*(1 + TX), equivalently

*ATX..         TX =E= PCA*POLINT / PL;

MODEL ALGEBRAIC /PRF_X.X, PRF_Y.Y, PRF_W.W,
                  MKT_X.PX, MKT_Y.PY, MKT_CA.PCA, MKT_L.PL,
                  MKT_W.PW,I_CONS.CONS, PPOL.POL, ATX.TX /;
```

```
*       Chose a numeraire: real consumer price index

PW.FX = 1;

*       Set initial values of variables:

X.L=1; Y.L=1; W.L=1; PX.L=1; PY.L=1; PL.L=1; POL.L = 1; PCA.L = 1;
CONS.L=300;

ALGEBRAIC.ITERLIM = 0;
SOLVE ALGEBRAIC USING MCP;
ALGEBRAIC.ITERLIM = 1000;
SOLVE ALGEBRAIC USING MCP;

WELOPT = 100*W.L;
TAXOPT = TX.L;
DISPLAY WELOPT, TAXOPT;

POLINT = 1.5;

SOLVE ALGEBRAIC USING MCP;

WELOPT = 100*W.L;
TAXOPT = TX.L;
DISPLAY WELOPT, TAXOPT;
```

6.7 Two households with different preferences, endowments

This model is an adaptation of model M3-7:

Here we introduce a social welfare function and find taxes that maximize social welfare.

This is modeled as an MPEC. Model M3-7 is the constraint set on the MPEC.

The model allows for variable (and endogenous) weights on each household type in social welfare. This model can be viewed as a basic starting point for thinking about political economy.

Two household: differ in preferences and in endowments

Household A: well endowed with labor,
preference for labor-int good Y
Household B: well endowed with capital,
preference for capital-int good X

	Production Sectors				Consumers	
Markets	X	Y	WA	WB	A	B
PX	100		-40	-60		
PY		100	-60	-40		
PWA			100		-100	
PWB				100		-100
PL	-25	-75			90	10
PK	-75	-25			10	90

Allows for tax to be redistributed unevenly between households, and for a lump-sum redistribution for comparison purposes.

The tax redistribution or sharing rule can also be interpreted as the relative number of households in each group, with all households getting an equal share of tax receipts.

Parameters

```
WEIGHTA  weight of consumer A in social welfare
WEIGHTB  weight of consumer B in social welfare
```

Variables

```
SHA      share of tax redistributed to consumer A
SHB      share of tax redistributed to consumer B
LS       lump sum redistribution to consumer A;
```

Add a variable, social welfare WS, and an equation giving the social welfare function.

```
OBJ..      WS =E= (WA**WEIGHTA) * (WB**WEIGHTB);
```

The income-balance constraints for the two consumer types reflect their redistributive shares of total tax revenue, and/or lump-sum redistribution.

```
I_CONSA.. CONSA =E= 90*PL + 10*PK +
                    SHA*TAX*100*X*PX/(1+TAX) + LS;

I_CONSB.. CONSB =E= 10*PL + 90*PK +
                    SHB*TAX*100*X*PX/(1+TAX) - LS;
```

To make the problem interesting, we (initially) have a single tax instrument, and production/consumption tax on X.

Thus the only available tax is distortionary and creates an aggregate welfare loss. There is a cost to redistributing income.

Note that the optimal tax might be a subsidy, so the variable TAX is specified as a free (unbounded) variable.

The model is calibrated so that, if the welfare weights on the two consumer groups are equal, the optimal tax is zero.

Now give a higher weight to households A: WEIGHTA = 0.7.

Perhaps 70% of all households by count (and votes) are type A

A higher weight on households A, will mean a positive tax for two reinforcing reasons:

Good X is capital intensive

And

Households B are capital abundant

And

Households B have a high preference for X in consumption

Thus a positive tax hurts households B and helps households A.

Experiment 1: maximize welfare holding shares SHA and SHB fixed at 0.5, and fixing LS at 0: no lump-sum transfers possible.

Experiment 2: maximize welfare allowing for endogenous shares

Experiment 3: maximize welfare allowing for lump-sum transfers

Experiment 4: reverse the weights on the two household types

```
$TITLE:  M6-7.GMS: two households with different preferences, endowments

$ONTEXT
 adaptation of model M3-7:
 distortionary tax can be used for redistribution
 modeled as an MPEC: find the optimal tax maximizing social welfare
 two add-ons
 (1)allows the redistributive shares of tax revenue to be endogenous
 (2) allows an optimal lump-sum redistribution for comparison

Two household: differ in preferences and in endowments
Household A: well endowed with labor,
  preference for labor-int good Y
Household B: well endowed with capital,
  preference for capital-int good X

                  Production Sectors                Consumers
   Markets   |    X       Y       WA       WB  |      A       B
   ----------------------------------------------------------
        PX   |  100              -40      -60  |
        PY   |          100      -60      -40  |
        PWA  |                   100           |  -100
        PWB  |                            100  |          -100
        PL   |  -25     -75                    |    90      10
        PK   |  -75     -25                    |    10      90
   ----------------------------------------------------------
```

```
The tax redistribution or sharing rule can also be interpreted
as the relative number of households in each group, with all
households getting an equal share of tax receipts
$OFFTEXT

PARAMETERS
        WEIGHTA  weight of consumer A in social welfare
        WEIGHTB  weight of consumer B in social welfere;

WEIGHTA = 0.5;
WEIGHTB = 0.5;


VARIABLES
        WS      social welfare
        TAX     endogenous tax rate on X
        LS      lump sum redistibution;

NONNEGATIVE VARIABLES

        X       Activity level for sector X,
        Y       Activity level for sector Y,
        WA      Activity level for weflare for consumer A
        WB      Activity level for welfare for consumer B
        PX      Price index for commodity X,
```

```
        PY       Price index for commodity Y,
        PK       Price index for primary factor K,
        PL       Price index for primary factor L,
        PWA      Price index for welfare A(expenditure function),
        PWB      Price index for welfare B(expenditure function),

        CONSA    Income definition for CONSA,
        CONSB    Income definition for CONSB

        SHA      share of tax redistributed to consumer A
        SHB      share of tax redistributed to consumer B;

EQUATIONS

        OBJ      Social welfare function
        PRF_X    Zero profit for sector X
        PRF_Y    Zero profit for sector Y
        PRF_WA   Zero profit for sector WA (Hicksian welfare index)
        PRF_WB   Zero profit for sector WB (Hicksian welfare index)

        MKT_X    Supply-demand balance for commodity X
        MKT_Y    Supply-demand balance for commodity Y
        MKT_L    Supply-demand balance for primary factor L
        MKT_K    Supply-demand balance for primary factor K
        MKT_WA   Supply-demand balance for aggregate demand consumer A
        MKT_WB   Supply-demand balance for aggregate demand consumer B
```

```
        I_CONSA Income definition for CONSA
        I_CONSB Income definition for CONSB
        ADDUP   Sum of the redistributive shares equals 1;

*       Objective function (social weflare function) to be maxmized

OBJ..           WS =E= (WA**WEIGHTA) * (WB**WEIGHTB);

*       Zero profit conditions:

PRF_X..         100 * (PL**0.25 * PK**0.75) * (1+TAX) =E= 100 * PX;

PRF_Y..         100 * (PL**0.75 * PK**0.25) =E= 100 * PY;

PRF_WA..        100 * PX**0.4 * PY**0.6 =E= 100 * PWA;

PRF_WB..        100 * PX**0.6 * PY**0.4 =E= 100 * PWB;

*       Market clearing conditions:

MKT_X..         100 * X =E= 40*WA*PWA/PX + 60*WB*PWB/PX;

MKT_Y..         100 * Y =E= 60*WA*PWA/PY + 40*WB*PWB/PY;

MKT_WA..        100 * WA =E= CONSA / PWA;
```

```
MKT_WB..          100 * WB =E= CONSB / PWB;

MKT_L..           90 + 10 =E= 25*X*(PX/(1+TAX))/PL + 75*Y*PY/PL;

MKT_K..           10 + 90 =E= 75*X*(PX/(1+TAX))/PK + 25*Y*PY/PK;

*       Income constraints:

I_CONSA..         CONSA =E= 90*PL + 10*PK + SHA*TAX*100*X*PX/(1+TAX) + LS;

I_CONSB..         CONSB =E= 10*PL + 90*PK + SHB*TAX*100*X*PX/(1+TAX) - LS;

ADDUP..           SHA + SHB =E= 1;

*MODEL MPEC /ALL/;

OPTION MPEC = nlpec;
MODEL MPEC /OBJ, PRF_X.X, PRF_Y.Y, PRF_WA.WA, PRF_WB.WB,
                 MKT_X.PX, MKT_Y.PY, MKT_L.PL,
                 MKT_K.PK, MKT_WA.PWA, MKT_WB.PWB,
                 I_CONSA.CONSA, I_CONSB.CONSB, ADDUP /;

*       Check the benchmark:
WS.L    =1;
X.L     =1;
```

```
Y.L      =1;
WA.L     =1;
WB.L     =1;

PL.L     =1;
PX.L     =1;
PY.L     =1;
PK.L     =1;
PWB.L    =1;
PWA.L    =1;

CONSA.L  =100;
CONSB.L  =100;

TAX.L     =0.;
SHA.L     =0.5;
SHB.L     =0.5;

PWA.FX = 1;

SOLVE MPEC USING MPEC MAXIMIZING WS;


* now allow weights in social welfare to differ
* e.g., 70% of all households/voters are type A
```

```
WEIGHTA = 0.7;
WEIGHTB = 0.3;

* first, fix shares at 0.5, hold LS = 0
SHA.FX = 0.5;
SHB.FX = 0.5;
LS.FX = 0;

SOLVE MPEC USING MPEC MAXIMIZING WS;

* now free up the redistributive weights

SHA.UP = +INF;
SHB.UP = +INF;
SHA.LO = 0;
SHB.LO = 0;

SOLVE MPEC USING MPEC MAXIMIZING WS;

* now allow lump-sum transfers

LS.UP = +INF;
LS.LO = -INF;

SOLVE MPEC USING MPEC MAXIMIZING WS;
```

```
* now switch the weights to consumer B

WEIGHTA = 0.3;
WEIGHTB = 0.7;

SOLVE MPEC USING MPEC MAXIMIZING WS;
```